TD2 - Courbes paramétrées

Nicolas Estibals*

4, 11 et 18 décembre 2008

Exercice 1 Quelques courbes à tracer

Il faudra utiliser les commandes `plot` et `polarplot`. Consultez l'aide pour comprendre leurs utilisations. Tracez les courbes définies par :

1. $\begin{cases} x(t) &= \cos^3(t) \\ y(t) &= \sin^3(t) \end{cases}$
2. $\begin{cases} x(t) &= \int_0^t \cos(\frac{x^2\pi}{2})\,dx \\ y(t) &= \int_0^t \sin(\frac{x^2\pi}{2})\,dx \end{cases}$
3. $\rho = \cos(\theta - \pi/3)$
4. $\rho = \cos(9\theta)$

Exercice 2 L'âne et la carotte

Un âne se déplace le long d'une ellipse de grand axe 200 m et de petit axe 50 m. Nous cherchons à trouver la trajectoire de la carotte qui pend au bout d'un bâton c m devant.

1. Rappelez l'équation paramétrique d'une telle ellipse centrée en $(0, 0)$.
2. Donnez les coordonnées du vecteur tangent normalisé au point de paramètre t.
3. Trouvez l'équation de la position de la carotte.
4. Tracez sur le même graphe la trajectoire de l'âne et de la carotte pour différentes longueurs de bâton.

Exercice 3 Eadem mutata resurgo

Nous allons étudier la courbe paramétrée définie par $\rho = \Phi^{\theta/\pi}$ ($\Phi = \frac{1+\sqrt{5}}{2}$).

1. Tracez cette courbe.
2. Déterminez la pente de la tangente à la courbe en un point $M(\rho, \theta)$.
3. Calculez l'angle entre la tangente et la droite (OM) et justifiez ainsi le nom de spirale équiangle pour cette courbe.
4. Nous allons maintenant justifier la maxime *eadem mutata resurgo* ("je renais changé à l'identique"). Montrer que la courbe est invariante par similitude d'angle β et de rapport $\Phi^{\beta/\pi}$.

Exercice 4 La roulette

La roulette est le nom donné au XVII$^{\text{ième}}$ siècle à la cycloïde, *id est* la courbe décrite par le point d'un cercle lorsque celui-ci roule sans glisser. Cette courbe peut être définie par :

$$\begin{cases} x(t) &= t - \sin(t) \\ y(t) &= 1 - cos(t) \end{cases}.$$

1. Tracez la courbe (oui, oui, encore tracer...).
2. Écrivez une fonction qui renvoie le cercle de rayon r et de centre (x, y).
3. Écrivez une fonction qui renvoie le rayon de centre $(t, 1)$ de longueur 1 qui passe par la cycloïde.
4. Montrez la construction de la courbe sur un graphe.

*N'hésitez pas à m'écrire à l'adresse `Nicolas.Estibals@ens-lyon.fr`!
Vous pouvez retrouver les sujets sur `http://perso.ens-lyon.fr/nicolas.estibals/`.

Exercice 1 : Quelques courbes à tracer

```
> restart; with(plots): # Nous allons beaucoup utiliser 'plot' ...
Warning, the name changecoords has been redefined
```

Question 1/

```
> plot([cos(t)^3,sin(t)^3,t=0..2*Pi],title="L'astéroïde");
```



Question 2/

```
> plot([int(cos(Pi*x**2/2),x=0..t),int(sin(Pi*x**2/2),x=0..t),t=0..5],ti
  le="Spirale de Cornu");
```



Question 3/

```
> polarplot(cos(t-Pi/3),t=0..2*Pi,title="Une paramétrisation du cercle")
```



Question 4/

```
> polarplot(cos(9*t),t=0..2*Pi,title="Marguerite ?");
```



Exercice 2 : L'âne et la carotte

```
> restart; with(plots):
```

Warning, the name changecoords has been redefined

Question 1/

```
> a:=200; b:=50;
  x:=t->a*cos(t);
  y:=t->b*sin(t);
  plot([x(t),y(t),t=0..2*Pi],title="La trajectoire de
  l'âne",scaling=constrained);
  # scaling = constrained permet d'imposer de respecter la même échelle
  pour
  #l'axe des abscisses et des ordonnées.
```

$$a := 200$$

$$b := 50$$

$$x := t \to a\cos(t)$$

$$y := t \to b \sin(t)$$

La trajectoire de l'âne



Question 2/

Il s'agit de dériver la position sur la courbe et de normaliser le vecteur obtenu. Nous allons définir une fonction de normalisation pour faciliter l'opération.

```
> normalise:=(x,y)->(x/sqrt(x**2+y**2),y/sqrt(x**2+y**2));
```

$$normalise := (x, y) \to \frac{x}{\sqrt{x^2 + y^2}}, \frac{y}{\sqrt{x^2 + y^2}}$$

```
> normalise(diff(x(t),t),diff(y(t),t));
```

$$-\frac{4 \sin(t)}{\sqrt{16 \sin(t)^2 + \cos(t)^2}}, \frac{\cos(t)}{\sqrt{16 \sin(t)^2 + \cos(t)^2}}$$

Question 3/

La carotte se trouve c mètres devant l'âne, il faut utiliser le vecteur précédent pour connaître la direction de l'âne.

```
> X:=t->x(t)+c*(-4*sin(t)/(16*sin(t)^2+cos(t)^2)^(1/2));
Y:=t->y(t)+c*(cos(t)/(16*sin(t)^2+cos(t)^2)^(1/2));
```

$$X := t \to x(t) - \frac{4\, c \sin(t)}{\sqrt{16 \sin(t)^2 + \cos(t)^2}}$$

$$Y := t \to y(t) + \frac{c \cos(t)}{\sqrt{16 \sin(t)^2 + \cos(t)^2}}$$

Question 4/

Il s'agit de bien construire l'ensemble (écrit entre {}) à tracer.

```
> plot({[x(t),y(t),t=0..2*Pi],
        seq([X(t),Y(t),t=0..2*Pi],c in {1,2, 5, 30})},
      title="L'âne et les carottes...",
      scaling=constrained);
```

L'âne et les carottes...



Exercice 3 : Eadem mutata resurgo

```
> restart;
```

Question 1/

La commande 'polarplot' permet de tracer une courbe définie en polaire.

```
> with(plots):
  polarplot( ((1+sqrt(5))/2)^(theta/Pi),theta=-10..10,title="Spirale logarithmique",scaling=constrained);
```

Warning, the name changecoords has been redefined

Spirale logarithmique



Question 2/

Pour calculer la pente de la tangente au point M il est plus simple de prendre l'équation paramérique de la courbe :

```
> x:=theta->((1+sqrt(5))/2)^(theta/Pi)*cos(theta);
  y:=theta->((1+sqrt(5))/2)^(theta/Pi)*sin(theta);
```

$$x := \theta \to \left(\frac{1}{2}+\frac{1}{2}\sqrt{5}\right)^{\left(\frac{\theta}{\pi}\right)} \cos(\theta)$$

$$y := \theta \to \left(\frac{1}{2}+\frac{1}{2}\sqrt{5}\right)^{\left(\frac{\theta}{\pi}\right)} \sin(\theta)$$

La pente de la tangente est alors y'/x'.

```
> m:=combine(diff(y(theta),theta)/diff(x(theta),theta));
  # 'combine' marche mieux que 'simplify' pour sdimplifier des expressions trigonométriques.
```

$$m := \frac{\sin(\theta)\ln\left(\frac{2}{1+\sqrt{5}}\right) - \cos(\theta)\,\pi}{\cos(\theta)\ln\left(\frac{2}{1+\sqrt{5}}\right) + \sin(\theta)\,\pi}$$

Question 3/

Pour connaître la pente dans le repère de Frenet, il faut calculer rho'(theta)/rho(theta).

```
> diff(((1+sqrt(5))/2)^(theta/Pi),theta)/((1+sqrt(5))/2)^(theta/Pi);
```

$$\frac{\ln\left(\frac{1}{2}+\frac{1}{2}\sqrt{5}\right)}{\pi}$$

Cette valeur est indépendante de theta, la courbe a donc un angle constant par rapport à (OM) (ou u_rho) qui est égale à :

```
> alpha:=arctan(%);
```

$$\alpha := \arctan\left(\frac{\ln\left(\frac{1}{2}+\frac{1}{2}\sqrt{5}\right)}{\pi}\right)$$

Question 4/

La similitude donnée en question s'écrit en polaire ainsi :

```
> simi:=(rho,theta)->(rho*((1+sqrt(5))/2)^(beta/Pi),theta+beta);
```

$$simi := (rho, theta) \to \rho\left(\frac{1}{2}+\frac{1}{2}\sqrt{5}\right)^{\left(\frac{\beta}{\pi}\right)}, \theta+\beta$$

Vérifions maintenant que la courbe est invariante par cette application :

```
> simi(((1+sqrt(5))/2)^(theta/Pi),theta);
```

$$\left(\frac{1}{2}+\frac{1}{2}\sqrt{5}\right)^{\left(\frac{\theta}{\pi}\right)} \left(\frac{1}{2}+\frac{1}{2}\sqrt{5}\right)^{\left(\frac{\beta}{\pi}\right)}, \theta+\beta$$

Cette équation est bien la même que celle du début de l'exercice. (Remplacer theta+beta dans |R par t dans |R)

Exercice 4 : La roulette

```
> restart; with(plots);
Warning, the name changecoords has been redefined
```

[*animate, animate3d, animatecurve, arrow, changecoords, complexplot, complexplot3d, conformal, conformal3d, contourplot, contourplot3d, coordplot, coordplot3d, cylinderplot, densityplot, display, display3d, fieldplot, fieldplot3d, gradplot, gradplot3d, graphplot3d, implicitplot, implicitplot3d, inequal, interactive, interactiveparams, listcontplot, listcontplot3d, listdensityplot, listplot, listplot3d, loglogplot, logplot, matrixplot, multiple, odeplot, pareto, plotcompare, pointplot, pointplot3d, polarplot, polygonplot, polygonplot3d, polyhedra_supported, polyhedraplot, replot, rootlocus, semilogplot, setoptions, setoptions3d, spacecurve, sparsematrixplot, sphereplot, surfdata, textplot, textplot3d, tubeplot*]

Question 1/

```
> x:=t->t-sin(t):
  y:=t->1-cos(t):
  plot([x(t),y(t),t=0..3*Pi],title="La  cycloïde",scaling=constrained);
```



Question 2/

Nous pourrons l'afficher après grâce à 'plot'.

```
> cercle:= (x,y,r) -> [x+r*cos(theta), y+r*sin(theta),theta=0..2*Pi];
```

$$cercle := (x, y, r) \rightarrow [x + r\cos(\theta), y + r\sin(\theta), \theta = 0 .. 2\pi]$$

Question 3/

```
> rayon:= t -> [x(t) + (t-x(t))*T,y(t)+ (1-y(t))*T,T=0..1];
```

$$rayon := t \rightarrow [x(t) + (t - x(t))\,T, y(t) + (1 - y(t))\,T, T = 0 .. 1]$$

Question 4/

```
> plot({[x(t),y(t),t=0..4*Pi],
        seq(cercle(t,1,1),t in {0,Pi/4,3*Pi/4,7*Pi/4,3*Pi/2}),
        seq(rayon(t),t in {0,Pi/4,3*Pi/4,7*Pi/4,3*Pi/2})},
        scaling=constrained);
```


2
1,5
1
0,5
0
0
2
4
6
8
10
12